Tricycle Lights
Embedded System Design, Lab 6

Ben Lorenzetti

October 29, 2015

Contents



(a) Potentiometer Input (b) Output LEDs

Figure 1: I/O Circuit Diagrams from the 44–Pin Demo Board User's Manual

1 Objectives and Problem Description

1.1 Tricycle Lights

Blink two LEDs, representing left and right turn signals, depending on the position of a rotary potentiometer, which represents a steering wheel. The following conditions should be met.

1. Use the potentiometer and LEDs on the 44–Pin Demo Board.
2. Use the LED connected to RD7 for the left turn signal, and RD0 for right.
3. When wheel is in neutral position (wiper in middle of pot.), both turn indicator lights should be off.
4. When turned counterclockwise or clockwise from neutral position, the left or right LEDs should blink at a rate proportional to the angular displacement from neutral position.

2 Procedure

2.1 Potentiometer and LEDs on the 44–Pin Demo Board

In this lab, we will need to measure a potentiometer input (the steering wheel) and blink two LEDs for output. The 44–pin demo board for the PIC16F887 has all of this hardware on board. The potentiometer and LEDs are connected to pins as shown in the circuit diagram in figure 1.

According to the specifications, the LED at RD7 should be the left blinker and the LED at RD0 should be the right blinker.

2.2 PIC16F887 Analog to Digital Converter (ADC)

2.3 PIC16F887 Clock Sources

2.4 Implementation Flowchart

2.5 Delay Function

2.6 Linear Mapping

3 Expected Results

4 Experiment and Design Revisions

4.1 Command Line Assembly

My `.asm` source files were assembled on the command line so please do this if they don't compile nicely in the IDE. On Ubuntu, with the default MPLAB installation location, from the directory containting `lfsr.asm`, the commands are:

```
$ cp /opt/microchip/mplabx/v3.10/mpasmx/p16f887.inc ./p16f887.inc
$ /opt/microchip/mplabx/v3.10/mpasmx/mpasmx -p16f887 lfsr.asm
$ more lfsr.ERR
```

5 Observations



Figure 2: Demonstration of Tricycle Turn Lights

6 Discussion

7 Tricycle Lights Implementation Code

```
; tricycle-lights.asm
; Ben Lorenzetti
; Embedded Systems Design, Fall 2015

#include <p16f887.inc>
        __CONFIG        _CONFIG1, _LVP_OFF & _FCMEN_OFF & _IESO_OFF & _BOR_OFF
            & _CPD_OFF & _CP_OFF & _MCLRE_OFF & _PWRTE_ON & _WDT_OFF &
            _INTRC_OSC_NOCLKOUT
        __CONFIG        _CONFIG2, _WRT_OFF & _BOR21V

#define NEUTRAL_POS             0x80
#define INNER_DELAY_TIME        0x8F
#define MIDDLE_DELAY_TIME       0x0F
#define MINIMUM_HALF_PERIOD     0x06
#define OSC8_CHANNEL0_NOGO_ADON B'01000001'
#define LEFTJUSTIFY_VSS_VDD     B'00000000'
#define RESOLUTION_MASK         B'11111100'

;-------------------------Organize Program Memory-----------------------;
Reset_Vector
        ORG 0
        GOTO Initialize

Interupt_Vector
        ORG .4

;---------------------Allocate Static Variables-------------------------;
        cblock  0x20
        adc_result
        turn_signal
        delay_time
        outer_delay_counter
        middle_delay_counter
        inner_delay_counter
        endc

;----------Pause (INNER_DELAY * MIDDLE_DELAY * delay_time)----------;
Delay_Function
        MOVF    delay_time, W           ; copy delay_time to
        MOVWF   outer_delay_counter     ;   outer_delay_counter
        MOVLW   INNER_DELAY_TIME        ; initialize
        MOVWF   inner_delay_counter     ;   inner_delay_counter
        MOVLW   MIDDLE_DELAY_TIME       ; initialize
        MOVWF   middle_delay_counter    ;   middle_delay_counter
Inner_Loop
        DECFSZ  inner_delay_counter, f
        GOTO    Inner_Loop
        MOVLW   INNER_DELAY_TIME
        MOVWF   inner_delay_counter
Middle_Loop
        DECFSZ  middle_delay_counter, f
        GOTO    Inner_Loop
```

```
        MOVLW   MIDDLE_DELAY_TIME
        MOVWF   middle_delay_counter
Outer_Loop
        DECFSZ  outer_delay_counter, f
        GOTO    Inner_Loop
        RETURN

;--------------------Initialize Data Memory-----------------------------;
Initialize
        ;-------------- Initialize I/O ------------------------------------;
        BANKSEL TRISD           ; select Register Bank 1
        CLRF    TRISD           ; set all LED pins to output
        BANKSEL PORTD           ; back to Register Bank 0
        CLRF    PORTD           ; set all LED pins to low
        BANKSEL TRISA
        CLRF    TRISA           ; clear TRISA
        BSF     TRISA, RA0      ; set port A pin 0 to input
        ;--------------- Initialize ADC-------------------------------------;
        BANKSEL ADCON1
        MOVLW   LEFTJUSTIFY_VSS_VDD
        MOVWF   ADCON1          ; left justify result,
                ; use VSS and VDD for Vref- and Vref+
        BANKSEL ADCON0
        MOVLW   OSC8_CHANNEL0_NOGO_ADON
        MOVWF   ADCON0          ; ADC clock rate = Fosc/8,
                ; ADC input channel = 0, ADC on
        MOVLW   10
        MOVWF   delay_time      ; initialize delay_time
        CALL    Delay_Function  ; Pause to allow ADC to settle

;----------------Begin Main Program Loop-----------------------------;
Main
        ;------------- Measure Potentiostat Input ----------------------;
        BANKSEL ADCON0
        BSF     ADCON0, GO      ; start convertion
        BTFSC   ADCON0, GO      ; is converstion done?
        GOTO    $-1             ; go back to BTFSC instruction
        BANKSEL ADRESH
        MOVFW   ADRESH          ; store ADC result in W
        BANKSEL PORTA           ; go back to bank 0
        ;------- Calculate Angular Displacement from Neutral -------;
        MOVLW   RESOLUTION_MASK ; reduce number of steps by
        ANDWF   ADRESH, 1       ;   truncating lower bits in ADRESH
        MOVLW   NEUTRAL_POS
        SUBWF   ADRESH, 1       ; compute displacement from Neutral
        ; Z = 1 if ADRESH == NEUTRAL_POS; C = 1 if ADRESH >= NEUTRAL_POS
        ;---------------- Perform Conditional Logic --------------------;
        BTFSC   STATUS, Z       ; test zero flag, skip next if clear
        GOTO    Main            ; if (ADRESH == NEUTRAL_POS)
        BTFSS   STATUS, C       ; if (ADRESH < NEUTRAL_POS), invert
        COMF    ADRESH, F       ;   angular displacement
        MOVLW   1 << RD7        ; assume left turn (ADRESH < NEUTRAL)
        BTFSC   STATUS, C       ; if actually (ADRESH > NEUTRAL_POS),
        MOVLW   1 << RD0        ;   then fix it to be right (RD0)
```

```
        ;-------------------- Blink LEDs ----------------------------------;
        MOVWF   PORTD           ; turn on LED
        MOVLW   MINIMUM_HALF_PERIOD
        MOVWF   delay_time      ; keep LED on for fixed delay time
        CALL    Delay_Function  ;
        CLRF    PORTD           ; turn off LEDs
        MOVF    ADRESH, W       ; compute appropriate delay time
        SUBLW   NEUTRAL_POS + MINIMUM_HALF_PERIOD
        MOVWF   delay_time      ; (from angular displacement value)
        CALL    Delay_Function  ; delay
        ;---------- End of Main Function Loop ------------------------------;
        GOTO    Main
;------------------------------ End of File ---------------------------------;
        END
```